

## Capitolul 8

### Interfața cu sistemul de operare DOS

În acest capitol vor fi prezentate elementele de bază ale interacțiunii programelor scrise în ASM cu cele două componente software de bază dintr-un calculator personal, subsistemul BIOS și sistemul de operare DOS, fără a se prezenta o listă exhaustivă a funcțiilor de sistem sau BIOS. Asemenea liste de funcții sunt disponibile în manuale specifice ale sistemului de operare DOS.

#### 8.1 Componentele de bază ale sistemului DOS

Componentele sistemului de operare DOS sunt:

- **DOS-BIOS.** Această componentă este memorată pe disc într-un fișier cu numele IBMBIO.COM sau IBMIO.SYS sau IO.SYS, depinzând de furnizor. Ea conține driverele pentru dispozitivele CON (tastatură și display), PRN (imprimantă), AUX (interfață serială) și CLOCK (ceas de timp real). Deasemenea, modulul DOS-BIOS conține drivere pentru discurile flexibile și pentru discurile hard. Accesul la aceste drivere va implica apelul unor rutine din ROM-BIOS (porțiunea de BIOS aflată în ROM, care conține procedurile dependente de hardware).
- **DOS-Kernel (nucleul DOS).** Această componentă este memorată pe disc în fișierul IBMDOS.COM sau MSDOS.SYS, depinzând de furnizor și conține funcții de acces la fișiere, funcții de intrare/ieșire la nivel de caracter etc. Aceste funcții operează într-un mod independent de hardware (e vorba de funcțiile DOS apelate prin instrucțiuni INT 21H).
- **Interpretorul de comenzi (Shell) standard.** Este memorat pe disc în fișierul COMMAND.COM. Interpretorul este cel care afișează prompterul DOS la consolă, acceptând și executând comenzi de la tastatură. Interpretorul de comenzi este la rândul său compus din 3 părți:
  - ♦ **porțiunea rezidentă**, care conține handleri (rutine de tratare) pentru acțiunile CTRL-Break (CTRL-C) de la tastatură și pentru erori critice (erori la citirea sau scrierea la dispozitive periferice, cum ar fi discurile sau imprimanta). Erorile critice conduc la afișarea la consolă a unor mesaje de forma:  

Error on device ... Abort, Retry, Fail ?
  - ♦ **porțiunea tranzitorie**, care afișează prompter-ul la consolă (de exemplu C:\>), citește comanda de la consolă și o execută. Porțiunea tranzitorie conține comenzile DOS interne (de genul DIR, TYPE, COPY etc.). Când un program executabil își încheie execuția, controlul revine porțiunii rezidente a interpretorului de comenzi, care verifică o sumă de control, detectând astfel dacă s-a modificat cumva porțiunea tranzitorie din memorie. În caz afirmativ, porțiunea tranzitorie se reîncarcă de pe disc;
  - ♦ **rutina de inițializare**, care este încărcată în memorie o dată cu încărcarea sistemului DOS de pe disc. Când operația de încărcare s-a terminat, această componentă nu mai este necesară și spațiul respectiv de memorie va fi folosit în alte scopuri.

Comenzile acceptate de interpretorul de comenzi standard sunt de 3 tipuri:

- ♦ comenzi interne (cum ar fi COPY, DIR, RENAME etc.), care nu au corespondent în fișiere executabile pe disc;
- ♦ comenzi externe, conținute în fișiere disc de tip .EXE sau .COM, care sunt încărcate în memorie în zona TPA (zona de procese tranzitorii);
- ♦ fișiere de comenzi indirecte (.BAT), care sunt fișiere text ce conțin comenzi interne, externe sau alte invocări de fișiere de comenzi indirecte.

Interpretorul testează întâi dacă linia introdusă de la tastatură este o comandă internă. Dacă nu este, se caută un fișier cu extensia .COM, .EXE

sau .BAT (în această ordine), întâi în directorul curent, apoi în toate directoarele specificate în comanda PATH. Dacă nu se găsește nici un fișier adecvat, se afișează la consolă mesajul:

```
Bad command or file name
```

Dacă s-a identificat un fișier cu extensia .COM sau .EXE, interpretorul încarcă fișierul executabil în memorie utilizând funcția EXEC (4BH) a sistemului de operare și îi dă controlul. Funcția EXEC construiește în memorie un bloc special numit PSP (Program Segment Prefix), în zona de procese tranzitorii TPA, la prima adresă disponibilă. Se încarcă apoi programul executabil, imediat după PSP și se execută operațiile de relocare, adică se ajustează valorilor simbolilor care conțin referințe la memorie, conform cu poziția fizică din memorie unde a fost încărcat programul.

- **ROM-BIOS.** Această componentă este rezidentă în memoria ROM a calculatorului și conține partea de autotestare de la punerea sub tensiune (numită și POST, de la **Power-On Self Test**), programul primar de încărcare de pe disc (**bootstrap**), rutinele de intrare/ieșire de nivel scăzut (dependente de hardware) etc. În general, această componentă nu rezidă numai pe placa de bază a calculatorului, anumite porțiuni fiind localizate pe plachetele de interfețe furnizate de producător.

### 8.2 Încărcarea sistemului DOS


La punerea sub tensiune, procesorul începe să execute instrucțiuni de la adresa fixă 0FFFF0H (care este situată în memoria de tip ROM). La această adresă se găsește o instrucțiune JMP FAR la o porțiune de program din ROM-BIOS, numită **POST (Power-On Self Test)**. Acest cod execută teste hardware pentru unitatea centrală, memoria internă, dispozitivele periferice etc., înscriind totodată vectorii de întrerupere pentru întreruperile hardware externe.

Se trece apoi la execuția unei zone de cod numită **ROM-Bootstrap** (încărcător primar), care testează dacă există dischete în unitățile de disc flexibil. Dacă nu, se trece la examinarea discului hard. Se citește apoi primul sector de pe pista 0 a discului (fie de pe discul flexibil, fie de pe discul hard). Acest sector este numit sector de boot (**Boot Sector**) și conține un alt program încărcător (numit **Disk-Bootstrap**) precum și informații despre organizarea discului. Programul **Disk-Bootstrap** este citit în memorie la nivel fizic și i se dă controlul. Unele calculatoare de generație mai nouă permit specificarea ordinii de examinare a unităților de disc.

Trebuie remarcat că, în acest moment, sistemul de operare nu este încărcat în memorie, deci nu se dispune de nici o funcție DOS. Harta memoriei la acest moment este ilustrată în Figura 8.1.

Programul încărcător citește primul sector al directorului rădăcină, determinând dacă primele două fișiere din acest director sunt, în ordine: **IO.SYS** (sau **IBMBIO.COM** sau **IBMBIO.SYS**) și **MSDOS.SYS** (sau **IBMDOS.COM**), care conțin componentele **DOS-BIOS** și **DOS-Kernel** ale sistemului de operare. Dacă aceste fișiere nu sunt identificate, se afișează la consolă mesajul:


```
Non-System disk or disk error  
Replace and strike any key when ready
```



**Figura 8.1 Harta memoriei imediat după încărcarea programului Disk-Bootstrap**

Dacă fișierele sunt identificate corect, programul Disk-Bootstrap le citește în memorie și transferă controlul la punctul de start al modulului IO.SYS.

Componenta DOS-BIOS constă din două module: BIOS-ul propriu-zis și modulul SYSINIT. Harta memoriei din acest moment este ilustrată în Figura 8.2.




**Figura 8.2 Harta memoriei după încărcarea modulelor IO.SYS și MSDOS.SYS**

Încărcarea celor două fișiere se face la adrese mici de memorie, deoarece nu se cunoaște încă dimensiunea memoriei. Modulul SYSINIT, apelat de către partea de inițializare a BIOS-ului, determină dimensiunea memoriei continue și se auto-realocă (adică se mută) la o adresă mare de memorie (către coada memoriei). În continuare, modulul SYSINIT mută modulul MSDOS.SYS din locul unde a fost încărcat inițial în poziția sa finală din memorie, suprascriind

vechea copie a modulului SYSINIT (încărcat inițial la adrese mici) precum și alte porțiuni inițiale de cod conținute în IO.SYS și care nu mai sunt necesare.

Se dă apoi controlul porțiunii de inițializare a modului DOS-Kernel. Acesta inițializează tabelele interne și zonele de lucru ale sistemului DOS, setează vectorii de întrerupere de pe nivelele 20H...2FH (adică întreruperile soft specifice funcțiilor DOS) și inițializează driverele rezidente ale dispozitivelor periferice periferice (discuri, consolă etc.). Controlul revine apoi modulului SYSINIT.

În acest moment, sistemul DOS este operațional, deci modulul SYSINIT poate apela funcții DOS. SYSINIT încearcă să deschidă fișierul CONFIG.SYS, care conține diverse comenzi specifice prin care se poate configura sistemul. În acest fișier se pot specifica drivere suplimentare care să fie încărcate (așa-numitele drivere instalabile), se poate specifica numărul de buffere pentru disc, numărul maxim de fișiere care pot fi deschise simultan, numele fișierului care conține interpretorul de comenzi etc. Dacă fișierul CONFIG.SYS este identificat în directorul rădăcină, el este încărcat în memorie și se execută fiecare comandă specificată. Harta memoriei din acest moment este ilustrată în Figura 8.3.




**Figura 8.3 Harta memoriei înainte de încărcarea interpretorului de comenzi**

După ce toate driverele instalabile au fost încărcate, SYSINIT închide toate fișierele (dispozitivele) folosite și apoi redeschide dispozitivele consolă (CON), imprimantă (PRN) și interfață serială (AUX), asociindu-le handler-ele de fișiere 0 (standard input), 1 (standard output), 2 (standard error), 3 (standard list) și 4 (standard auxiliary). Handler-ele 0, 1, 2 sunt asignate la dispozitivul CON, handler-ul 3 este asignat la dispozitivul PRN iar handler-ul 4 la dispozitivul AUX. Aceste handler-e pot fi reasignate prin apeluri de funcții sistem adecvate.

În final, modulul SYSINIT apelează funcția EXEC a sistemului de operare, pentru a încărca și executa interpretorul standard de comenzi. Odată ce interpretorul a fost încărcat, el afișează la consolă prompterul specific și așteaptă introducerea de comenzi de la tastatură. Porțiunea tranzitorie din


COMMAND.COM este încarcată peste SYSINIT, care nu mai este necesar. În final, Imaginea memoriei este cea din Figura 8.4.



**Figura 8.4 Imaginea finală a memoriei**

### **8.3 Structura Blocului PSP (Program Segment Prefix)**

Blocul PSP este caracteristic ambelor categorii de programe executabile (.EXE și .COM), fiind o zonă de 256 de octeți care precede imaginea din memorie a programului executabil. La lansarea în execuție a unui program, adresele DS:0000 și ES:0000 indică începutul blocului PSP asociat programului. PSP-ul este construit de către funcția EXEC la încărcarea programului de pe disc. Structura PSP-ului este ilustrată în Figura 8.5.



**Figura 8.5 Structura blocului PSP**

Semnificația câmpurilor din Figura 8.5 este următoarea:

- *octeții 0-1* - conțin codul unei instrucțiuni INT 20H (ieșire în DOS), care permite unui program executabil să se încheie cu o instrucțiune JMP sau RET la adresa PSP:0; această tehnică nu mai este actuală, pentru terminarea unui program utilizându-se funcția DOS 4CH;
- *octeții 2-3* - conțin dimensiunea memoriei de bază continue (vârful memoriei), în paragrafe de câte 16 octeți); această informație este folosită în procesul de încărcare a programului executabil;
- *octetul 4* - rezervat;
- *octeții 5-9* - conțin codul unei instrucțiuni CALL de tip far la dispecerul de funcții DOS;
- *octeții 0AH-0DH* - conțin adresa far a rutinei de încheiere (accesibilă prin INT 22H); întreruperea 22H este o altă modalitate perimată de a încheia programele executabile; actualmente se utilizează funcția DOS 4CH;
- *octeții 0EH-10H* - conțin adresa far a rutinei de tratare a apăsării tastelor Ctrl-Break (accesibilă prin INT 23H); Ctrl-Break provoacă terminarea forțată a programelor în curs de execuție;
- *octeții 12H-15H* - conțin adresa far a rutinei de tratare a erorilor critice (accesibilă prin INT 23H); (erorile critice sunt de genul: nu există dischetă în unitatea de disc flexibil și se dă o comandă de genul DIR A:);
- *octeții 16H-2BH* - rezervați;
- *octeții 2CH-2DH* - conțin adresa de segment a blocului de environment; acesta este o zonă de memorie în format text, care conține informații de genul: căii (path) active, specificarea interpretorului de comenzi, specificarea prompterului de comenzi etc; între altele, în blocul de environment se găsește calea completă a fișierului care conține


programul executabil; în felul acesta, se poate ști în program numele fișierului executabil din care a fost încărcat programul; acesta este suportul fizic al argumentului argv[0] al funcției main de la programele C;

- octeții 2EH-5BH - rezervați;
- octeții 5CH-6BH și 6CH-7FH - reprezintă așa numiții parametri formatați; sunt utilizați pentru două blocuri de control fișiere (FCB-uri), care reprezintă o modalitate perimată de lucru cu fișiere disc; actualmente se utilizează funcții de prelucrare a fișierelor orientate pe handler-e (vezi 8. ???);
- octetul 80H - reprezintă lungimea (numărul de caractere) conținute în zona de parametri neformatați;
- octeții 81H-OFFH - reprezintă zona de parametri neformatați; această zonă conține caracterele introduse în linia de comandă după numele fișierului executabil (așa numita coadă a comenzii); prin această zonă, un program executabil are acces la parametrii din linia de comandă (acesta este suportul oferit de DOS pentru parametrii numiți generic argc și argv de la funcția main a programelor C); parametrii din linia de comandă nu cuprind paratrii de redirectare a dispozitivelor standard de intrare și de ieșire; primii 43 de octeți de la offsetul 80H mai sunt utilizați implicit și ca Zonă de Transfer Disc (DTA); această zonă este folosită de unele funcții de acces la disc și se poate poziționa într-un spațiu rezervat de către utilizator prin funcția DOS 1AH; de asemenea, adresa curentă a zonei DTA poate fi citită cu funcția DOS 2FH;

### 8.4 Structura programelor de tip COM

Programele de tip COM corespund modelului de memorie tiny, în care există un singur segment și toate salturile și apelurile de proceduri sunt de tip near. Fișierul disk conține exclusiv modulul care se încarcă, fără alte informații suplimentare, ca în cazul fișierelor de tip EXE.

Imaginea memoriei unui program COM este prezentată în Figura 8.6.



**Figura 8.6** Imaginea memoriei ocupate de un program COM

Programele COM au ca punct de start adresa near 100H, adică imediat după blocul PSP. Indicatorul SP este plasat la adresa OFFFEH, deci pe ultimul cuvânt adresabil. Acest cuvânt e inițializat cu 0.

Un program COM se poate încheia în unul din următoarele moduri:

- prin apel al funcției standard de terminare (funcția DOS 4CH)
- prin apel al întreruperii soft INT 20H
- prin execuția unei instrucțiuni RETN, dacă registrul SP conține valoarea inițială OFFFEH. Cuvântul de la adresa OFFFEH fiind 0, instrucțiunea RETN va da controlul la adresa near 0, adică la primul octet din PSP. Acolo însă se găsește codul unei instrucțiuni INT 20H, care provoacă revenirea în DOS.

## Gheorghe Muscă – Programare in Limbaj de Asamblare

Programele COM nu necesită relocare, deoarece toți simbolii relativi sunt deplasamente în cadrul unui segment unic. Simpla încărcare a registrului CS este suficientă.

Sablonul de dezvoltare pentru programele COM este:

```

    _TEXT      segment para 'CODE'
    ORG        100H
    ASSUME     CS:_TEXT, DS:_TEXT, ES:_TEXT, SS:_TEXT
start:
    jmp      init
    ;
    ; Aici se pun eventuale definiții de date
    ;
init:
    ;
    ; Aici se pune codul propriu-zis
    ;
    mov     ax, 4C00H           ; Apel funcție DOS
    int     21H                ; pentru terminare program
_TEXT ENDS
end start                       ; Etichetă de start
```

Programele de tip COM (ca de altfel și cele de tip EXE) își rezervă toată memoria disponibilă. În unele situații e necesară reducerea la minim a memoriei rezervate, care se poate realiza printr-un apel al funcției DOS 4AH. Șablonul de dezvoltare al programelor COM va fi următorul:

```

    _TEXT      segment para 'CODE'
    ORG        100H
    ASSUME     CS:_TEXT, DS:_TEXT, ES:_TEXT, SS:_TEXT
start:
    jmp      init
    ;
    ; Aici se pun definiții de date
    ;
main proc near
    ;
    ; Aici se pune codul propriu-zis
    ;
    mov     ax, 4C00H
    int     21H
    main    endp
    ;
    ; Aici se pun alte proceduri (daca e cazul)
    ;
init:
    mov     ah, 4AH             ; Funcție redimensionare memorie
    mov     bx, offset end_mem  ; Limita pina unde se pastreaza
    mov     cl, 4               ; Dimensiunea memoriei se
    shr     bx, cl              ; specifică în paragrafe de 16 octeți
    inc     bx                   ; Mai punem unul de siguranță
    int     21H                 ; Apel propriu-zis
    mov     sp, offset end_mem  ; Inițializare sp
    jmp     main
    ; Rezervăm o stivă de 512 octeți, care să cuprindă și spațiul dintre etichetele
    ; init și init_mem, zonă ce se poate suprascrie după încărcarea programului
init_mem:
    db (512 - (init_mem - init)) dup (?)
    ;
```

```

; Până aici rezervăm memoria
;
end_mem:
_TEXT ENDS
end start

```

## 8.5 Structura și încărcarea fișierelor EXE

Programele de tip EXE pot fi încărcate în orice locație de memorie și se compun din mai multe segmente. Acest fapt impune ajustarea tuturor instrucțiunilor care conțin adrese de segment, cum ar fi:

- JMP/CALL FAR PTR nume
- MOV reg, SEG nume;
- MOV reg\_seg, valoare.

Acest proces se numește **relocare** și se execută la încărcarea programului în memorie. Punctul de start poate fi oarecare și este stabilit prin directiva END.

Este evident că un fișier de tip EXE trebuie să conțină, pe lângă programul propriu-zis, și informații despre simbolii relocabile, adresa de start, adresa segmentului de stivă etc. Aceste informații sunt grupate în prima parte a fișierului, numită **header al fișierului EXE**. Structura acestui header este ilustrată în Tabelul 8.7.

Offset	Dim.	Conținut	Semnificație
0	2	4DH 5AH	Semnătură fișier EXE ('MZ')
2	2	PartPag	Lungime fișier modulo 512
4	2	PageCnt	Lungime în pagini de 512 octeți, inclusiv header
6	2	ReloCnt	Număr de elemente din tabela de relocare
8	2	HdrSize	Dimensiune header în paragrafe de 16 octeți
0AH	2	MinMem	Necesar minim de memorie peste sfârșitul programului (în paragrafe)
0CH	2	MaxMem	Necesar maxim de memorie peste sfârșitul programului (în paragrafe)
0EH	2	ReloSS	Deplasament segment stivă
10H	2	ExeSP	Valoare SP la execuție
12H	2	ChkSum	Sumă de control
14H	2	ExeIP	Valoare IP (adresă de start)
16H	2	ReloCS	Deplasament segment de cod
18H	2	Tabloff	Poziția în fișier a tabelii de relocare (uzual 1CH)
1AH	2	Overlay	Indicator overlay (0 pentru modulele de bază)
?	4*	offs   seg	Tabela

		...	...	de
		offs	seg	relocare
?	?			Caractere până la limita de paragraf

**Tabelul 8.7 Structura header-ului fișierelor .EXE**

Primii doi octeți identifică tipul fișierului. Următorii 4 octeți conțin lungimea fișierului (inclusiv header-ul), în pagini de 512 octeți și, respectiv, modulo 512. Suma de control reprezintă suma cu semn schimbat a tuturor cuvintelor din fișier, permițând un control al validității fișierului.

Valorile **ReloSS** și **ReloCS** reprezintă deplasamentele segmentelor de stivă și cod față de adresa de segment de încărcare (segmentele pot fi în orice ordine). Valorile **ExeSP** și **ExeIP** reprezintă conținutul registrelor **SP** și **IP** la intrare în execuție. Aceste valori sunt deduse din dimensionarea segmentului de stivă și din directiva END a modulului de program principal (care precizează eticheta de start).

Tabela de relocare cuprinde adresele tuturor cuvintelor care trebuie ajustate, adrese precizate prin offset în cadrul segmentului curent și prin deplasamentul segmentului respectiv față de adresa de segment inițială. Tabela de relocare are **ReloCnt** (vezi [6]) elemente, începe la poziția **Tabloff** (vezi [18H]) în fișier și ocupă **ReloCnt\*4** octeți. Header-ul este completat cu octeți fără semnificație, până la o limită de paragraf.

Relocarea programului este realizată de funcția **EXEC** a sistemului de operare (funcția 4BH) și constă în următorul algoritm:

- Se creează un PSP cu funcția DOS 26H;
- Se citesc 1CH octeți din fișierul .EXE (așa-numita porțiune formatată a header-ului), într-o zonă locală de memorie;
- Se determină lungimea modulului = ((PageCnt \* 512) - (HdrSize \* 16)) - PartPag;
- Se determină deplasamentul în fișier al modulului = (HdrSize \* 16);
- Se selectează o adresă de segment START\_SEG (uzual PSP + 10H);
- Se citește modulul de program în memorie la adresa START\_SEG:0000;
- Se setează poziția de citire din fișier la începutul tabelii de relocare (Tabloff);
- Pentru fiecare element al tabelii de relocare (ReloCnt):
  - ◆ se citește elementul din tabela ca 2 cuvinte (I\_OFF, I\_SEG);
  - ◆ se determină adresa actuală de memorie a elementului care trebuie relocat RELO\_SEG = (START\_SEG + I\_SEG);
  - ◆ se citește elementul care trebuie relocat, deci cuvântul de la adresa (RELO\_SEG:I\_OFF);
  - ◆ se adună START\_SEG la acest cuvânt (se ajustează adresa de segment);
  - ◆ se depune valoarea ajustată înapoi la adresa (RELO\_SEG:I\_OFF);
- Se alocă memorie suplimentară pentru program, conform MaxMem și MinMem;
- Se inițializează registrele semnificative și se dă controlul programului:
  - ◆ ES = DS = PSP;
  - ◆ SS = START\_SEG + ReloSS;
  - ◆ SP = ExeSP;
  - ◆ CS = START\_SEG + ReloCS;
  - ◆ IP = ExeIP.

Poziționarea registrelor CS și IP se realizează print-o secvență de forma:

```
PUSH  START_SEG + ReloCS  
PUSH  ExeIP  
RETF                               ; Intrare în execuție
```

## 8.6 Programe de tip TSR (Terminate and Stay Resident)

Programele de tip EXE sau COM obișnuite sunt încărcate în memorie în zona de procese tranzitorii (TPA) și ocupă memoria până la încheierea execuției. Mai concret spus, la execuția funcției DOS 4CH, memoria ocupată de program este eliberată și oferită sistemului DOS.

Programele de tip TSR (care sunt tot programe de tip EXE sau COM) rămân în memorie o durată nedeterminată de timp (sunt **rezidente**). Ele sunt **activate** de către evenimente externe, cum ar fi acționarea unei taste special destinate (**hot-key**) sau expirarea unui interval de timp. Pentru ca să rămână rezident, un program trebuie să-și încheie execuția cu un apel al funcției DOS 31H, în care se precizează (în registrul DX) numărul de paragrafe de memorie care se păstrează rezidente.

Se deosebesc deci următoarele noțiuni:

- **activarea unui program TSR** - reprezintă întreruperea programului care se rulează în mod curent (nu neapărat un program utilizator), cu salvarea corespunzătoare a întregului context și transferul controlului către programul TSR;
- **dezactivarea unui program TSR** - reprezintă refacerea contextului programului întrerupt și transferul controlului către acesta;
- **instalarea unui program TSR** - reprezintă încărcarea de pe disc și încheierea execuției prin apelul funcției 31H;
- **dezinstalarea unui program TSR** - reprezintă eliberarea memoriei ocupate de un program TSR instalat anterior.

### 8.6.1 Activarea și dezactivarea programelor TSR

Activarea se face de obicei ca urmare a acționării unei taste special destinate. Se utilizează întreruperea pe nivelul 9 (asociat tastaturii), care se produce la fiecare apăsare și ridicare a unei taste.

Interacțiunea cu întreruperea 9 se face prin modificarea ("furtul") vectorului de întrerupere cu o rutină proprie de tratare, care apelează rutina veche, testând dacă s-a apăsât tasta specială de activare a TSR-ului. Rutina veche trebuie apelată, deoarece ea convertește codul de scanare produs de tastatură într-un cod ASCII, actualizează buffer-ul tastaturii, gestionează starea tastelor Ctrl, Shift, Alt etc. De asemenea, este posibil să fie instalate mai multe TSR-uri "unul peste altul", activate prin taste speciale diferite, care au modificat la rândul lor vectorul 9 de întrerupere. Prin apelul rutinei vechi, se iau în considerare și aceste TSR-uri instalate anterior.

Apelul rutinei vechi se face simulând o instrucțiune INT, prin secvența:

```
pushf  
call  dword ptr cs:oldint
```

unde oldint este o locație în care s-a salvat adresa vechii rutine de tratare.

Trebuie observat faptul esențial că, la activarea unui TSR, dintre registrele de segment nu se poate conta decât pe conținutul registrului CS. Acesta este poziționat în urma invocării rutinei de tratare pe nivelul 9, în timp ce restul

registrelor de segment sunt poziționate conform programului care a fost întrerupt. De aceea, programele TSR se dezvoltă de obicei sub forma unui singur segment, adresabil prin registrul CS. Aceasta explică prezența prefixului de segment din instrucțiunea CALL de mai sus. Dacă este cazul, se pot încărca și alte registre de segment cu valori adecvate, dar aceste valori trebuie memorate la etapa de instalare, obligatoriu în locații din segmentul adresabil prin registrul CS.

Secvența de parcurgere a mai multor programe TSR instalate este ilustrată în Figura 8.8.




Figura 8.8 Parcurgerea succesivă a mai multor rutine de tratare pe nivelul 9

### Non-reenranța funcțiilor DOS

Problemele speciale legate de activarea TSR-urilor sunt în principal datorate faptului că sistemul DOS **nu este reentrant**. Aceasta înseamnă că, dacă se întrerupe execuția unei funcții DOS și din rutina de întrerupere se apelează o altă funcție DOS, este posibil ca funcționarea sistemului de operare să fie compromisă.

Soluția constă în activarea TSR-urilor numai în situații "sigure", adică, teoretic, atunci când nu se execută funcții DOS. Se utilizează o combinație de două tehnici:

- Prima tehnică se bazează pe un flag de sistem, numit uzual flag INDOS, care menține un contor al nivelurilor de adâncime ale apelurilor de funcții DOS. Adresa flagului INDOS poate fi obținută prin funcția DOS 34H. Teoretic, ar trebui să activăm TSR-ul doar atunci când acest contor este nul. Din păcate, flagul INDOS este diferit de zero mai tot timpul, datorită interpretorului de comenzi care așteaptă introducerea comenzilor de la tastatură prin execuția funcției DOS 0AH (citire șir de caractere).

- A doua tehnică se bazează pe întreruperea soft 28H, care este apelată periodic de către sistem atunci când interpretorul așteaptă comenzi de la consolă. Se consideră că, în momentul apelului acestei întreruperi, este relativ sigur să se activeze programul TSR.

În loc de a activa programul TSR imediat după ce a apărut cererea de activare, se va memora această cerere într-o variabilă, urmând ca ulterior, când sunt îndeplinite condițiile de activare, să se facă activarea propriu-zisă. Concret, se va modifica rutina de tratare pe nivelul 28H, astfel încât să se testeze variabila care memorează cererea de activare, și, dacă există cerere, să se activeze TSR-ul.

### **Acțiuni critice în timp**

O altă categorie de probleme o reprezintă **acțiunile critice în timp**. Acestea sunt activități ale sistemului de operare sau ale BIOS-ului, a căror întrerupere ar putea duce la funcționări defectuoase. Concret, este vorba despre rutinele de nivel scăzut pentru operațiile cu discurile, realizate prin întreruperea de BIOS 13H și de operațiile de nivel scăzut asupra display-ului, realizate prin întreruperea BIOS 10H.

Problema se rezolvă prin modificarea întreruperilor 13H și 10H, adăugându-se poziționarea la 1 a unui flag propriu, pe durata cât aceste rutine de tratare sunt active. Programul TSR trebuie activat doar dacă flagul respectiv este nul. Structura noilor rutine de tratare va fi:

```
new13 proc far
    mov flag_bios, 1 ; Marchează operații critice în curs
    pushf ; Simulare execuție
    call dword ptr cs:old13 ; apel INT 13H
    mov flag_bios, 0 ; Marchează sfârșit operații critice
    retf 2 ; Revenire cu flagurile întoarse de
    ; apelul lui old13
new13 endp
```

În care old13 este o variabilă DWORD care memorează adresa vechii rutine de tratare, iar flag\_bios este o variabilă rezervată în segmentul adresabil prin CS. Se observă încheierea procedurii prin RETF 2 și nu prin IRET (deși este procedură de tratare a unei întreruperi), ceea ce menține flagurile așa cum au fost întoarse de apelul vechii rutine de pe nivelul 13H.

### **Evitarea activării recursive**

O problemă de bază la implementarea programelor TSR o reprezintă **evitarea activării recursive**. Dacă programul TSR a fost activat și se află în execuție, este posibil să apară o nouă cerere de activare (de exemplu, se apasă din nou tasta specială de activare). În această situație, programul TSR nu trebuie reactivat. Problema se rezolvă prin menținerea unui alt flag propriu, care este 1 pe durata cât TSR-ul este activ și 0 în rest.

### **Comutarea contextului la activare și dezactivare**

Programele TSR pot fi destul de complexe, în sensul că pot apela orice funcție DOS, lucra cu fișiere disc etc. Aceasta presupune ca activarea să fie însoțită de o salvare cât mai completă a contextului programului întrerupt și de o comutare a contextului către programul TSR. Similar, ldezactivarea TSR-ului, trebuie refăcut complet contextul programului întrerupt.

Problemele care trebuie avute în vedere sunt următoarele:

## Gheorghe Muscă – Programare in Limbaj de Asamblare

- salvarea și restaurarea registrelor programului întrerupt;
- salvarea și restaurarea unor structuri de date specifice programului întrerupt și comutarea pe structurile specifice programului TSR:
  - ♦ salvarea adresei PSP-ului programului întrerupt și marcarea ca fiind activ a PSP-ului programului TSR;
  - ♦ salvarea zonei DTA și comutarea pe zona DTA a TSR-ului;
- comutarea pe stiva proprie a programului TSR;
- salvarea unei porțiuni (uzual 64 de octeți) din stiva programului întrerupt într-o zonă proprie;
- salvarea conținutului ecranului.

Toate operațiile de mai sus trebuie realizate în sens invers la cedare controlului către programul întrerupt.

Pentru citirea și marcarea PSP-ului și a zonei DTA sunt disponibile funcții DOS specifice. Salvarea unei porțiuni din stiva programului întrerupt este necesară deoarece sistemul DOS lucrează cu stive proprii alocate la adrese fixe și este posibil ca întreruperea să fi apărut atunci când registrele SS:SP erau poziționate pe o asemenea stivă. Un apel al unei alte funcții DOS va utiliza poate aceeași stivă DOS, ceea ce va distruge conținutul stivei apelului anterior.

Salvarea și restaurarea ecranului apare ca necesară în situația în care TSR-ul afișează ceva la consolă (ceea ce se întâmplă destul de frecvent). Salvarea presupune un transfer între memoria video și o zonă proprie. De obicei, programele TSR lucrează în mod text, deci se va utiliza memoria video de la adresa 0B0000H (la ecrane monocrom) și 0B8000H la ecrane color. Funcție de modul video curent, se salvează 25 sau 43 de linii de ecran, fiecare a 80 de cuvinte (un simbol pe ecran este codificat prin caracterul ASCII extins și printr-un octet de atribut, care precizează culoarea sau efecte speciale de afișare). Este indicat să se salveze și modul video curent precum și poziția cursorului pe ecran. Dacă este instalat un driver de mouse, trebuie salvată poziția cursorului mousului și inhibată afișarea mousului pe durata cât TSR-ul este activ. Dacă este cazul, se poate comuta pe o rutină (handler) proprie de tratare a evenimentelor de la mouse.

Pentru citirea și setarea cursoarelor, a modurilor video etc., se utilizează funcții BIOS specifice.

### 8.6.2 Instalarea și dezinstalarea programelor TSR


Instalarea unui program TSR înseamnă încărcarea lui în memorie și pregătirea condițiilor pentru activare. Dezinstalarea unui TSR înseamnă eliberarea memoriei ocupate la instalare. Trebuie remarcat că dezinstalarea trebuie făcută de către un alt program decât cel rezident (eventual o altă instanță a aceluiași program).

#### *Evitarea instalării repetate*

Programul TSR este memorat într-un fișier executabil. Nimic nu ne împiedică să lansăm în execuție acest program de mai multe ori. Totuși, nu se poate accepta ca un TSR să fie instalat de mai multe ori, deoarece s-ar consuma în mod inutil memorie. Pe de altă parte, secvența de activare din Figura 8.8 ar fi compromisă (care instanță se va activa ?)

Un program TSR trebuie deci să testeze dacă nu este instalat deja în memorie și să refuze instalarea în acest caz.

Testarea se face definind o secvență particulară de caractere (o semnătură) la un offset fix în cadrul unicului segment care compune programul TSR. Identificarea locului în care programul TSR este eventual rezident se poate face numai prin examinarea vectorului de întrerupere care declanșează activarea (uzual, INT 9). Citind acest vector, identificăm adresa de segment a rutinei de tratare, adică adresa de segment a programului TSR care este eventual instalat. Comparând semnătura programului curent cu semnătura din segmentul indicat de vectorul INT 9, se determină dacă TSR-ul este deja instalat (vezi Figura 8.9).



**Figura 8.9 Tehnica de testarea a instalării**

Această tehnică este posibilă numai dacă TSR-ul deja instalat este ultimul din lanțul de TSR-uri care au modificat vectorul de întrerupere pe nivelul 9.

O altă soluție, mai generală dar mai lentă, este parcurgerea tuturor blocurilor de memorie alocate de către sistem. Aceste blocuri sunt structuri de date specifice DOS care conțin informații despre deținătorul blocului, deci cărui modul de program i-a fost alocat blocul respectiv. Astfel, se caută de fapt în memorie numele fișierului executabil care conține programul TSR. În acest capitol, vom utiliza prima tehnică de identificare.

Instalarea trebuie să salveze o serie de date necesare procesului de activare, cum ar fi adresa de PSP și a zonei DTA a programului TSR, adresa stivei programului TSR etc. De asemenea, trebuie rezervat spațiu pentru stiva TSR, pentru salvarea ecranului, pentru copiile vectorilor de întrerupere modificați etc.

Instalarea TSR-ului mai presupune și modificarea vectorilor de întrerupere 9, 10H, 13H etc. cu adresele propriilor rutine de tratare. În final, se calculează dimensiune spațiului de memorie care trebuie să rămână rezident și se încheie execuția cu apelul funcției DOS 31H.

**Dezinstalarea TSR-urilor**


La identificarea situație de TSR deja instalat, multe programe se mulțumesc să afișeze la consolă un mesaj de genul:

```
Program XXXXX already installed!
```

și apoi să-și încheie execuția în mod normal.

O variantă mai evoluată este ca, la a doua invocare a programului TSR, să se realizeze dezinstalarea primei instanțe, după care a doua instanță să se încheie cu ieșire normală în DOS.

Dezinstalarea trebuie să elibereze spațiul de memorie alocat primei instanțe. Trebuie mai întâi eliberat blocul de environment (mediu) al TSR-ului instalat. Adresa blocului de environment se găsește la un offset fix în cadrul PSP-ului (vezi 8.3). La instalare, este necesară memorarea adresei blocului PSP într-o variabilă aflată la un offset fix. După ce s-a eliberat blocul de environment, se eliberează memoria propriu-zisă cu funcția 49H și se încheie execuția cu funcția DOS 4CH. Figura 8.10 ilustrează acest proces.



**Figura 8.10 Tehnica de dezinstalare**

De observat că, funcție de testul asupra instalării TSR-ului, se va face instalarea instanței curente sau dezinstalarea instanței anterior instalate. Tot la

dezinstalare se refac vectorii 9, 10H, 13H și 28H, care au fost modificați la instalare.

### 8.6.3 Șablon de dezvoltare pentru programe TSR

Următorul program reprezintă un șablon de dezvoltare pentru aplicațiile TSR uzuale, fiecare acțiune fiind comentată intensiv. Trebuie observat accesul la buffer-ul tastaturii, realizat prin pointerii de la adresele absolute 400:1AH și 400:1CH, care indică coada cele două capete ale buffer-ului. Buffer-ul tastaturii se află în spațiul de adrese 400:1EH-400:3CH, fiind organizat circular. Buffer-ul este vid când cei doi pointeri sunt egali și este plin când pointerul de introducere este cu o poziție (2 octeți) în urma pointerului de extragere. Capacitatea logică este de 30 de octeți (15 taste) iar cea fizică de 32 de octeți (16 taste).

Acest model de program realizează doar ștergerea ecranului, afișarea unui mesaj și așteptarea apăsării unei taste oarecare. Pentru afișări la consolă se folosesc direct funcții DOS, ceea ce explică terminatorul '\$' pentru șiruri de caractere. Tasta de activare este F1. La o a doua execuție, programul se dezinstalează.

Se observă și funcțiile DOS sau BIOS pentru citirea sau poziționarea diverselor date de sistem.

```
    bios segment at 40h                ; Segment date BIOS
        org    1AH
        b_out  dw    ?                ; Indicator extragere buffer BIOS
        b_in   dw    ?                ; Indicator introducere buffer BIOS
    bios ends

    cseg segment
    assume cs:cseg
    MONO equ    0
    F1    equ    3B00H                ; 59 în zecimal = cod F1
    HOTKEY equ    F1                  ; Tastă activare

    ident    db    '_IDENTIF_'
    lng      equ    $ - ident
    mes_act  db    'TSR activ: Apasati orice tasta '
            db    'pentru a reveni'
            db    7, '$'
    mes_inst db    ' TSR instalat: Activare cu F1', '$'
    mes_dezinst db    ' TSR dezinstalat', '$'
;
;
;
    old9    label dword
            old9_off    dw    ?
            old9_seg    dw    ?
    old10   label dword
            old10_off   dw    ?
            old10_seg   dw    ?
    old13   label dword
            old13_off   dw    ?
            old13_seg   dw    ?
    old28   label dword
            old28_off   dw    ?
            old28_seg   dw    ?
```

```

;
;   Spațiu pentru adresa flagului INDOS
;
flag_dos      label  dword
indos_off    dw      ?
indos_seg     dw      ?
;
flag_tsr      dw      0           ; Flag TSR activ
flag_bios     dw      0           ; Flag BIOS activ
cerere        db      0           ; Memorează cerere viitoare
cod_scan      dw      ?           ; Cod tastă activare
ss_int        dw      ?           ; Salvare registre stivă
sp_int        dw      ?           ; program întrerupt
ss_tsr        dw      cseg        ; Stivă locală TSR
              dw      256 dup (?)
sp_tsr        dw      sp_tsr
buff_video    db      4000 dup (?) ; Buffer salvare ecran
salv_cursor   dw      ?           ; Salvare cursor
dta           db      43 dup (?)   ; DTA local TSR
psp_tsr       dw      ?           ; Adresă PSP TSR
dta_tsr       dd      dta         ; Adresă DTA TSR
;
psp_int       dw      ?           ; Salvare PSP și
dta_int       dd      ?           ; DTA program întrerupt
;
;   Noile rutine de tratare întreruperi 9, 10H, 13H, 28H
;
new10 proc far
    mov     flag_bios, 1
    pushf
    call    dword ptr cs:old10
    mov     flag_bios, 0
    retf    2
new10 endp
new13 proc far
    mov     flag_bios, 1
    pushf
    call    dword ptr cs:old13
    mov     flag_bios, 0
    retf    2
new13 endp
new9  proc far
    assume ds:bios
    pushf
    call    cs:old9           ; Apel rutină veche
    cli                    ; Dezactivare întreruperi
    cmp     cs:flag_tsr, 0   ; Test deja activ
    jne     new9_end
    push    ds               ; Salvări
    push    bx               ; registre
    push    ax
    mov     ax, bios
    mov     ds, ax          ; Acces BIOS prin ds
    mov     bx, b_out       ; Indicator extr. buffer BIOS
new9_0:
    cmp     bx, b_in        ; Este buffer-ul BIOS vid ?
    je      new9_1         ; Da, return
    mov     ax, [bx]        ; Nu e vid, iau codul

```

## Gheorghe Muscă – Programare in Limbaj de Asamblare

```
    mov    cs:cod_scan, ax           ; și îl memorez
    cmp    ax, HOTKEY               ; Este tastă de activare ?
    jnz    new9_1                   ; Nu, return
    add    bx, 2                    ; Incrementez (extrag din buffer)
    cmp    bx, 3EH                  ; Compar cu val. maxima + 2
    jne    new9_11                  ; Nu este = 3EH
    mov    bx, 1eh                  ; Este 3EH și se aduce la val. inițiala
new9_11:
    mov    b_out, bx                ; Memorare indicator buffer BIOS
    jmp    new9_2
new9_1:
    pop    ax
    pop    bx
    pop    ds
    iret

assume    ds: NOTHING

new9_2:
    pop    ax
    pop    bx
    pop    ds

    cmp    cs:flag_bios, 0          ; Test operații critice
    je     cont1
    mov    cs:cerere, 1            ; Marchează cerere
new9_end:
    iret                            ; Revenire în programul întrerupt
cont1:
    push   ds
    push   bx
    lds    bx, cs:flag_dos         ; Test INDOS
    cmp    byte ptr [bx], 0
    je     cont2
    mov    cs:cerere, 1            ; Marchează cerere
    pop    bx
    pop    ds
    iret                            ; Revenire în programul întrerupt
cont2:
    pop    bx
    pop    ds
    call   activ_tsr              ; Apel funcție TSR
    iret
new9     endp
;
new28   proc far
    pushf                          ; Apel rutină
    call  dword ptr cs:old28        ; veche
    cli
    cmp   flag_tsr, 0              ; Test TSR activ
    jz    cont3
    iret
cont3:
    cmp   flag_bios, 0             ; Test operații critice
    jz    cont4
cont5:
    ret   2
cont4:
    cmp   cerere, 0                ; Este cerere ?
```

```

        jz     cont5
        mov   cerere,0
        call  near ptr activ_tsr           ; Satisface cererea
        ret   2
new28 endp
;
;     Rutina de baza care realizează activarea TSR-ului
;
activ_tsr proc near
        cli
        mov   flag_tsr, 1                 ; Marchează TSR activ
;
;     Comutarea contextului
;
        mov   sp_int, sp                  ; Salvează stiva
        mov   sp, ss                       ; programului întrerupt
        mov   ss_int, sp
        mov   sp, ss_tsr                   ; Comută pe stiva TSR-ului
        mov   ss, sp
        mov   sp, sp_tsr
;
        push  ax                           ; Salvări
        push  bx                           ; registre
        push  cx
        push  dx
        push  si
        push  di
        push  bp
        push  ds
        push  es
;
        mov   cx, 64                       ; Salvare 64
        mov   ax, ss_int                   ; de octeți
        mov   ds, ax                       ; din stiva
        mov   si, sp_int                   ; programului
        cld                                 ; întrerupt
iar1:
        push  [si]                          ; în stiva
        inc  si                              ; proprie
        inc  si
        loop iar1
;
        mov   ah, 62H                       ; Citește PSP-ul
        int   21h                           ; programului întrerupt
        mov   psp_int, bx                   ; și salvează
;
        push  es
        mov   ah, 2FH                       ; Citește DTA
        int   21h                           ; programului întrerupt
        mov   word ptr dta_int, bx          ; și
        mov   word ptr dta_int+2, es       ; salvează
        pop   es
;
        mov   ah,50H                         ; Marchează PSP-ul
        mov   bx,psp_tsr                    ; TSR-ului
        int   21h                           ; ca fiind cel activ
;
        push  ds                             ; Comută
        mov   ah,1ah                         ; pe zona DTA

```



```

        mov     ah, 2
        int     10H
;
        mov     ax, cs                ; Refacere
        mov     ds, ax                ; ecran
        mov     si, offset buff_video
;
IF MONO
        mov     ax, 0B000H
ELSE
        mov     ax, 0B800H
ENDIF
;
        mov     es, ax
        xor     di, di
        cld
        mov     cx, 4000
        rep     movsb
;
end_tsr:
        cli
        mov     ah, 50H                ; Comutare pe PSP-ul
        mov     bx, psp_int            ; programului întrerupt
        int     21H
;
        push    ds
        mov     ah, 1aH                ; Comutare pe DTA a
        lds     dx, dta_int            ; programului întrerupt
        int     21H
        pop     ds
;
        mov     cx, 64                ; Refacere conținut stivă
        mov     ax, ss_int              ; program
        mov     ds, ax                ; întrerupt
        mov     si, sp_int              ; din stiva
        add     si, 128                 ; curentă
iar2:
        dec     si
        dec     si
        pop     [si]
        loop   iar2
;
        pop     es                    ; Refacere
        pop     ds                    ; registre
        pop     bp                    ; din stivă
        pop     di                    ; curenta (TSR)
        pop     si
        pop     dx
        pop     cx
        pop     bx
        pop     ax
;
        mov     sp, ss_int              ; Refacere registre stivă
        mov     ss, sp                ; ale programului
        mov     sp, sp_int              ; întrerupt
;
        mov     flag_tsr, 0            ; Marcare TSR inactiv
        sti
        ret

```

```
activ_tsr endp
;
; Rutina urmatoare reface vectorii de intrerupre
; și eliberează memoria alocată pentru programul TSR
;
dezinstdproc near
;
; În ES : segmentul programului rezident
;
push ds
push es
cli
;
; Refacere vectori de intrerupere 9, 10H, 13H, 28H
;
mov ax, 2509H ; Vector 9
mov ds, es:old9_seg
mov dx, es:old9_off
int 21H
;
mov ax, 2510H ; Vector 10H
mov ds, es:old10_seg
mov dx, es:old10_off
int 21H
;
mov ax, 2513H ; Vector 13H
mov ds, es:old13_seg
mov dx, es:old13_off
int 21H
;
mov ax, 2528H ; Vector 28H
mov ds, es:old28_seg
mov dx, es:old28_off
int 21H
sti
;
; Prima dată eliberăm blocul de environment, a cărui adresă
; de segment se găsește la offset-ul 2CH în PSP
;
mov bx, es:psp_tsr
mov es, bx ; Adresă PSP
mov es, es:[2ch] ; ES = adresă bloc environment
mov ah, 49H ; Eliberare memorie alocată
int 21H ; la instalare
;
; Acum eliberăm memoria deținută de programul
; TSR, începând de la PSP
;
mov es, bx ; ES = PSP-ul TSR-ului
mov ah, 49H ; Eliberare
int 21H
;
pop es
pop ds
ret ; Revenire
dezinstdendp
;
sfirsit label byte ; Etichetă pentru a păstra codul
; rezident numai până aici
```

```

;
;
;   Aici e punctul de start al programului
;
start:
    mov  ax, cseg                ; Inițializări registre
    mov  ds, ax                  ; de
    mov  ss, ax                  ; segment
    mov  sp, sp_tsr              ; și stivă
;
;
;   Testăm dacă TSR-ul este deja instalat
;
    mov  ax, 3509H                ; Citire vector de
    int  21H                      ; întrerupere 9
                                ; ES = segment program instalat
                                ; deja (eventual)

    mov  ax, cs
    mov  ds, ax                  ; DS = segment program curent
    xor  si, si                  ; Offseturi 0
    mov  di, si
    mov  cx, lng                  ; Lungime mesaj
    repz cmpsb                   ; Comparăm ds:[si] cu es:[di]
    jnz  not_inst                ; Nu e instalat, salt
;
    call dezinst                 ; Deinstalare, apoi
    mov  dx, offset mes_dezinst  ; mesaj deinstalare
    mov  ah, 9
    int  21H
;
    mov  ax, 4c00H                ; Ieșire normală în DOS
    int  21H

not_inst:
;
;   Instalare TSR
;
    mov  ah, 62H                  ; Citire PSP și salvare
    int  21H                      ; pentru comutare context
    mov  psp_tsr, bx              ; la activări viitoare
;
    mov  ah, 1aH                  ; Similar pentru DTA
    lds  dx, dta_tsr
    int  21H
;
;   Citiri și salvări vectori vechi
;   (nivelele 9, 10H, 13H, 28H)
;
    mov  ax, 3509H
    int  21H
    mov  cs:old9_off, bx
    mov  cs:old9_seg, es          ; Vector 9
    mov  ax, 3510H
    int  21H
    mov  cs:old10_off, bx
    mov  cs:old10_seg, es         ; Vector 10H
    mov  ax, 3513H
    int  21H
    mov  cs:old13_off, bx
    mov  cs:old13_seg, es         ; Vector 13H
    mov  ax, 3528H

```

## Gheorghe Muscă – Programare in Limbaj de Asamblare

```
int 21H
mov cs:old28_off, bx
mov cs:old28_seg, es ; Vector 28H
mov ax, cs
mov ds, ax ; DS = segmentul de cod
;
push ds
push es
;
;
; Poziționare vectori noi
;
mov ax, 2509H
mov dx, offset new9
int 21H ; Vector 9 nou
;
mov ax, 2510H
mov dx, offset new10
int 21H ; Vector 10H nou
;
mov ax, 2513H
mov dx, offset new13
int 21H ; Vector 13H nou
;
mov ax, 2528H
mov dx, offset new28
int 21H ; Vector 28H nou
;
; Citire și salvare adresă flag INDOS
;
mov ax, 3400H
int 21H
mov cs:indos_off, bx
mov cs:indos_seg, es
;
pop es
pop ds
;
mov ah, 9
mov dx, offset mes_inst ; Mesaj instalare
int 21H
;
; Calcul acoperitor al necesarului de memorie (inclusiv PSP-ul)
;
mov dx, offset sfirsit + 100h
mov cl, 4 ; Spațiul se specifică
shr dx, cl ; în paragrafe de câte 16 octeți
inc dx
mov ax, 3100H ; Terminare cu rămânere în memorie
int 21H
cseg ends
end start
```

## 8.7 Funcții de intrare/ieșire orientate pe handleri

Handlerele sunt variabile întregi mai mari sau egale cu zero, întoarse de sistemul DOS sau predefinite, care identifică în mod univoc fișiere disc sau dispozitive. Funcțiile orientate pe handleri au generalitate maximă și permit schimbarea dispozitivelor de intrare și/sau ieșire asociate unui program într-o manieră foarte comodă.

Sunt predefinite următoarele handleri:

- 0 - numit și **stdin** (**Standard Input- Intrare Standard**), uzual asignat la dispozitivul consolă;
- 1- numit și **stdout** (**Standard Output - Ieșire Standard**), uzual asignat la dispozitivul consolă;
- 2 - numit și **stderr** (**Standard Error - Ieșire standard pentru mesaje de eroare**), totdeauna asignat la consolă;
- 3 - numit și **stdaux** (**Standard Auxilliary - Dispozitiv auxiliar standard**), uzual asignat la dispozitivul COM1;
- 4 - numit și **stdprn** (**Standard Printer - Imprimantă standard**), uzual asignat la dispozitivul LPT1.

Funcțiile orientate pe handleri pot avea acces la orice subdirector al discurilor, pornind de la rădăcină sau de la directorul curent.

Intrarea și ieșirea standard asociate unui program executabil pot fi redirectate din linia de comandă, cu ajutorul caracterelor < (redirectare **stdin**) și > (redirectare **stdout**). Dacă un program executabil **nume.exe** utilizează funcții de intrare/ieșire orientate pe handleri, atunci o lansare în execuție de forma:

```
C:\> nume < file1
```

va face ca toate citirile care au loc de la consolă să fie înlocuite cu citiri din fișierul **file1**.

Similar, o lansare în execuție de forma:

```
C:\> nume > file2
```

va face ca toate afișările care au loc la consolă să fie înlocuite cu scrieri în fișierul **file2**.

În fine, o linie de comandă de forma:

```
C:\> nume < file1 > file2
```

va redirecta atât **stdin** cât și **stdout**.

Interpretorul de comenzi permite și un simulacru de mecanism **pipe**, prin care ieșirea standard a unui program **nume1** este redirectată către intrarea standard a altui program **nume2**:

```
C:\> nume1 | nume2
```

Cum sistemul DOS nu poate executa programe în mod concurent, linia de mai sus este echivalentă cu secvența:

```
C:\> nume1 > temp  
C:\> nume2 < temp  
C:\> del temp
```

În continuare, sunt descrise principalele funcții de intrare/ieșire orientate pe handleri și parametrii acestora, respectiv valorile întoarse.

• **Create File (Creează sau recreează fișier)**

- ◆ Parametri de intrare:
  - AH = 3CH (cod funcție);
  - CX = attribute (Read Only, Hidden, System etc.);
  - DS:DX = adresă șir de caractere terminat cu 0 (calea fișierului).
- ◆ Parametri de ieșire:
  - CF = 0, AX = handler (dacă operația a fost încheiată cu succes);
  - CF = 1, AX = cod de eroare (eșec).
- ◆ Descriere:

Se creează fișierul cu numele specificat; dacă fișierul exista anterior, conținutul vechi se pierde.

• **Open File (Deschide fișier)**

- ◆ Parametri de intrare:
  - AH = 3DH (cod funcție);
  - AL = mod acces (0 = pentru citire, 1 = pentru scriere, 2 = punere la zi etc.);
  - DS:DX = adresă șir de caractere terminat cu 0 (calea fișierului).
- ◆ Parametri de ieșire:
  - CF = 0, AX = handler (dacă operați a fost încheiată cu succes);
  - CF = 1, AX = cod de eroare (eșec).
- ◆ Descriere:

Se deschide fișierul cu numele specificat, pentru operația specificată; fișierul trebuie să existe; la deschidere pentru scriere, conținutul vechi se pierde.

• **Read File (Citește din fișier)**

- ◆ Parametri de intrare:
  - AH = 3FH (cod funcție);
  - BX = handler (obținut la deschidere sau predefinit);
  - DS:DX = adresă buffer de citire (dimensionat corespunzător);
  - CX = număr de octeți cerut pentru citire.
- ◆ Parametri de ieșire:
  - CF = 0, AX = număr de octeți citați efectiv (dacă operația a fost încheiată cu succes);
  - CF = 1, AX = cod de eroare (eșec);
- ◆ Descriere:

Se încearcă citirea a CX octeți din fișier, din poziția curentă; poziția este avansată automat; dacă de la poziția curentă până la sfârșitul fișierului sunt mai puțin de CX octeti, se citesc câți octeți există (posibil 0); dacă AX = 0, s-a ajuns la EOF înainte de apelul funcției, iar dacă AX < CX, s-a ajuns la EOF în decursul execuției funcției; ambele situații sunt corecte.

• **Write File (Scrie în fișier)**

- ◆ Parametri de intrare:
  - AH = 40H (cod funcție);
  - BX = handler (obținut la deschidere sau predefinit);
  - DS:DX = adresă buffer de scriere;
  - CX = număr de octeți cerut pentru scriere.
- ◆ Parametri de ieșire:
  - CF = 0, AX = număr de octeți scriși efectiv (AX diferă de CX numai în situații de eroare);

CF = 1, AX = cod de eroare (eșec).

◆ Descriere:

Se încearcă scrierea a CX octeți în fișier, din poziția curentă; poziția este avansată automat; dacă CX < AX, este vorba de o situație de eroare.

### • Close File (Închide fișier)

◆ Parametri de intrare:

AH = 3EH (cod funcție);

BX = handler (obținut la deschidere).

◆ Parametri de ieșire:

CF = 0 (operație încheiată cu succes);

CF = 1, AX = cod de eroare (eșec).

◆ Descriere:

Se închide fișierul, adică se taie legătura logică dintre handle și fișier, stabilită la deschidere; operația este obligatorie, în special la fișierele deschise pentru scriere, deoarece acum se scriu bufferele interne ale sistemului DOS în fișier.

### • Delete File (Șterge fișier)

◆ Parametri de intrare:

AH = 41H (cod funcție);

DS:DX = adresă șir de caractere terminat cu 0 (calea fișierului).

◆ Parametri de ieșire:

CF = 0 (operație încheiată cu succes);

CF = 1, AX = cod de eroare (eșec).

◆ Descriere:

Se șterge fișierul cu numele specificat.

### • Duplicate Handle (Duplică handler)

◆ Parametri de intrare:

AH = 45H (cod funcție);

BX = handler (obținut la deschidere sau predefinit).

◆ Parametri de ieșire:

CF = 0, AX = handler duplicat (operație încheiată cu succes);

CF = 1, AX = cod de eroare (eșec).

◆ Descriere:

Se obține o copie a handlerului asociat fișierului; operația este necesară înainte de redirectarea unor fișiere sau dispozitive, în situația în care dorim și revenirea la asocierea inițială.

### • Redirect Handle (Redirectează handler)

◆ Parametri de intrare:

AH = 46H (cod funcție);

CX = handler care trebuie redirectat;

BX = handler (obținut la deschidere sau predefinit).

◆ Parametri de ieșire:

CF = 0, AX = handler redirectat (operație încheiată cu succes);

CF = 1, AX = cod de eroare (eșec);

◆ Descriere:

Redirectează handlerul CX către handlerul BX; handlerul BX a fost obținut în urma unei deschideri de fișier sau în urma unei duplicări.

• **Seek File (Poziționează indicatorul fișierului)**

- ◆ Parametri de intrare:
  - AH = 42H (cod funcție);
  - BX = handler (obținut la deschidere);
  - CX:DX = poziția cerută în fișier (întreg pe 32 de biți, CX = partea high, DX = partea low);
  - AL = punctul de referință (0=față de începutul fișierului, 1=față de poziția curentă, 2=față de sfârșitul fișierului).
- ◆ Parametri de ieșire:
  - CF = 0, DX:AX = noua poziție în fișier (operație încheiată cu succes);
  - CF = 1, AX = cod de eroare (eșec).
- ◆ Descriere:

Se poziționează indicatorul intern al fișierului într-o poziție absolută. Are sens la exploatarea în acces direct a fișierelor, în care dorim să controlăm explicit poziția din care se va citi sau în care se va scrie. De exemplu, după ce am făcut operații de citire, vrem să revenim la începutul fișierului, pentru o nouă operație de citire. Acest lucru se poate realiza cu CX=DX=0, AL=0.

• **Find First (Identifică primul fișier dintr-un nume generic cu \* și/sau ?)**

- ◆ Parametri de intrare:
  - AH = 4EH (cod funcție);
  - DS:DX = adresă șir de caractere terminat cu 0 (poate conține '\*' și '?');
  - CX = attribute (0 = normal).
- ◆ Parametri de ieșire:
  - CF = 0, zona DTA conține informații despre fișierul identificat (operație încheiată cu succes);
  - CF = 1, AX = cod eroare (nu s-a identificat fișier).
- ◆ Descriere:

Se caută primul fișier care verifică numele generic specificat. La offset-ul 1EH în zona DTA se găsește numele acestui fișier; este necesară obținerea adresei zonei DTA în prealabil (necesară la funcția următoare).

• **Find Next (Identifică următorul fișier dintr-un nume generic cu \* și/sau ?)**

- ◆ Parametri de intrare:
  - AH = 4FH (cod funcție);
  - DS:DX = adresa zonei DTA sau a unei copii a acestei zone.
- ◆ Parametri de ieșire:
  - CF = 0, zona DTA conține informații despre următorul fișier identificat (operație încheiată cu succes);
  - CF = 1, AX = cod eroare (nu s-a identificat fișier).
- ◆ Descriere:

Se caută următorul fișier care verifică numele generic specificat la un apel anterior al funcției Find First. La offset-ul 1EH în zona DTA se găsește numele acestui fișier; este necesară obținerea adresei zonei DTA în prealabil. În general, această funcție se apelează ciclic, până când se obține CF = 1 (nu mai sunt fișiere care să verifice numele generice).

## Gheorghe Muscă – Programare in Limbaj de Asamblare

Funcțiile de duplicare și de redirectare sunt utilizate de interpretorul de comenzi, pentru redirectarea din linia de comandă a intrării și ieșirii standard. Redirectări asemănătoare pot fi făcute și din programele utilizator. Să considerăm următorul program, care execută următoarele acțiuni:

- scrie un mesaj la ieșirea standard;
- creează un fișier disc;
- duplică handlerul 1 (ieșirea standard);
- redirectează ieșirea standard către fișierul anterior creat;
- scrie un mesaj la ieșirea standard (scrierea se va face de fapt în fișier);
- redirectează ieșirea standard la handlerul duplicat (revenire la situația inițială);
- scrie un mesaj la ieșirea standard.

Programul este elistat în continuare:

```
.model large
include io.h
.data
    nume          db    'fis.dat', 0
    h1            dw    ?
    h2            dw    ?
    mes_1         db    'Acest mesaj se va tipari la consola', cr, lf
    lng_1         dw    $ - mes_1
    mes_2         db    'Acest mesaj se va tipari in fis.dat', cr, lf
    lng_2         dw    $ - mes_2
    mes_3         db    'Acest mesaj se va tipari din nou la consola', cr, lf
    lng_3         dw    $ - mes_3
.stack 1024
.code
start:
    init_ds_es
;
    mov  ah, 40H                ; Scriere la
    mov  bx, 1                  ; standard output
    lea  dx, mes_1
    mov  cx, lng_1
    int  21H
;
    mov  ah, 45H                ; Duplicare
    mov  bx, 1                  ; standard output
    int  21H
    mov  h2, ax                 ; Handler duplicat in h2
;
    mov  ah, 3CH                ; Deschidere 'fis.dat'
    mov  cx, 0                  ; pentru scriere
    lea  dx, nume
    int  21H
    mov  h1, ax                 ; Handler in h1
;
    mov  ah, 46H                ; Redirecteaza
    mov  cx, 1                  ; standard output
    mov  bx, h1                 ; la h1, adica la 'fis.dat'
    int  21H
;
    mov  ah, 40H                ; Scriere la
    mov  bx, 1                  ; standard output redirectat
    lea  dx, mes_2
    mov  cx, lng_2
```

```
        int    21H
;
        mov    ah, 46H           ; Redirecteaza
        mov    cx, 1            ; standard output
        mov    bx, h2          ; la h2, adica la cel original
        int    21H
;
        mov    ah, 40H         ; Scriere la
        mov    bx, 1           ; standard output original
        lea    dx, mes_3
        mov    cx, lng_3
        int    21H
;
        exit_dos
end      start
```

Mesajele mes\_1 și mes\_3 vor fi afișate pe ecran, iar mesajul mes\_2 va fi scris în fișierul fis.dat. Programul de mai sus (să presupunem că se numește redi.exe) se poate acum rula și cu redirectare din linia de comandă:

```
C:\> redi > fis_nou.dat
```

ceea ce va duce la scrierea mesajelor mes\_1 și mes\_2 în fișierul fis\_nou.dat. Se observă că redirectările efectuate prin program se referă la asignarea curentă a dispozitivului respectiv, nu la asignarea implicită.

O problemă importantă la scrierea fișierelor executabile este accesul la parametrii din linia de comandă. La descrierea blocului PSP, am văzut că în PSP există un contor al caracterelor din linia de comandă, iar "coada" propriu-zisă a comenzii se găsește începând de la offsetul 81H.

Să considerăm un program care acceptă în linia de comandă un nume generic de fișier (cu '\*' sau '?') și afișează la consolă toate fișierele care corespund numelui generic. Accesul la fișiere se va face prin funcțiile Find First și Find Next, iar parametrii din linia de comandă se citesc din PSP.

Se pune problema corectitudinii numărului de parametri (cuvinte). Vom utiliza procedura count, dezvoltată în subcapitolul 8.9, care primește un șir de caractere și întoarce numărul de cuvinte din acel șir. Este necesară o ușoară modificare, în sensul de a accepta ca delimitatori de cuvinte numai spațiul și caracterul Tab. Pentru apelul funcțiilor DOS se utilizează macroinstrucțiunile definite în fișierul io.h (descriș în Anexa A).

Implementarea programului este următoarea:

```
.model    large
include io.h
        extrn count : far
.stack 512
;
;      Zonă proprie DTA
;
dta     struc
        db    30    dup ( )
        f_name db    13    dup ( )      ; Nume fișier identificat
dta     ends
;
        stdout equ 1
;
```

```

.data
file_struct    dta    <>
buf_name      db     128    dup (0)
gen_name      db     80     dup (0)
f_hand        dw     ?
nr_bytes      dw     ?
buf           db     4096   dup (?)
dim_buf       equ    $ - buf
help_err      db     cr, lf, 'Sintaxa este: dump <file>',0
mess_no_file  db     cr, lf, 'dump: Nu exista fisier: ',0
open_err      db     cr, lf, 'dump: Eroare deschidere fisier ',0
read_err      db     cr, lf, 'dump: Eroare citire fisier ',0

.code
start:
    cld
    push ds                ; Avem nevoie de adresa PSP
    pop  ax
    init_ds_es
    mov  si, 80H           ; Offset contor caractere
    lea  di, buf_name
    push ds
    mov  ds, ax
    mov  cl, [si]
    xor  ch, ch            ; CX = număr caractere, fără CR, LF
    inc  si                ; Primul caracter
    jcxz help_ds          ; Dacă CX = 0, mesaj de help
;
    push cx
    rep movsb              ; Transfer din PSP in buf_name
    mov  byte ptr es:[di], 0 ; Terminator șir
    pop  cx
    pop  ds
    lea  si, buf_name
ch_nxt:
    cmp  byte ptr [si], ' ' ; Sărim peste spațiile albe
    jz   skip
    cmp  byte ptr [si], tab ; și peste Tab-uri
    jz   skip
    jmp  first_ch
skip:
    inc  si                ; De la cap la coadă
    dec  cx                ; și contorizăm
    jmp  ch_nxt
;
first_ch:
    lea  di, gen_name      ; Transfer în gen_name (nume generic)
    jcxz help              ; Dacă acum CX = 0, nu există parametri
    rep movsb              ; Abia acum transfer
ch_last:
    dec  di                ; Pe ultimul caracter
    cmp  byte ptr [di], ' ' ; Eliminăm spațiile albe și de la sfârșit
    jz   skip1
    cmp  byte ptr [di], tab ; La fel cu Tab-urile
    jz   skip1
    jmp  last_char
skip1:
    dec  di                ; De la coadă la cap
    dec  cx                ; și contorizăm
    jmp  ch_last

```

```

last_char:
    inc    di                ; Am decrementat o dată în plus
    mov    byte ptr [di], 0    ; Punem terminator
    lea    si, gen_name        ; Adresă șir curățat de spații albe
                                ; inițiale și finale
    call   far ptr count        ; Numărăm cuvintele
    cmp    ax, 1                ; AX = numărul de cuvinte
    je     ok                    ; Dacă e unul singur, e bine
    jmp    help                  ; Altfel, mesaj help
;
help_ds:
    pop    ds                    ; Help cu POP DS
help:
    puts   help_err              ; Mesaj de asistență
    exit_dos                       ; și ieșire în DOS
;
ok:
    set_dta file_struct          ; Setăm DTA pe zona noastră
    find_first gen_name          ; Apel Find First
    jnc    rel                    ; S-a găsit ceva
    jmp    gata_1                 ; Nu s-a găsit nimic
rel:
    o_read file_struct.f_name, f_hand ; Deschide fișierul cu numele
                                ; raportat de Find First în DTA
    jnc    no_err                ; Test eroare
    puts   open_err              ; Mesaj de eroare
    puts   file_struct.f_name    ; Afișare nume fișier
    putsi  <cr, lf>              ; CR, LF
    jmp    next                  ; Și trecem la următorul fișier
no_err:
    putsi  <cr, lf, 'Fișier: '> ; Nume
    puts   file_struct.f_name    ; fișier identificat
    putsi  <cr, lf, cr ,lf>
iar:
    f_read f_hand, buf, dim_buf   ; Citire dim_buf octeți din fișier
                                ; în bufferul buf
    jc     error                  ; Test eroare
    test   ax, ax                ; Test sfârșit de fișier (s-au citit
                                ; 0 octeți)
    jz     next                  ; Salt la următorul fișier
    mov    nr_bytes, ax          ; Memorăm cât s-a citit de fapt
    f_writestdout, buf, nr_bytes ; Afișăm ce sa- citit
    cmp    nr_bytes, dim_buf     ; Dacă s-au citit mai puțini octeți
    jb     close                 ; decât am cerut, salt la închidere
    jmp    iar                   ; Altfel, reluăm bucla de citire
close:
    f_closef_hand                ; Închidere fișier
next:
    find_next file_struct         ; Următorul fișier
    jc     gata                  ; CF = 1 : nu mai sunt
    jmp    rel
error:
    puts   read_err              ; Meaj de eroare
    puts   file_struct.f_name    ; Nume fișier
    putsi  <cr, lf>
    jmp    next                  ; Următorul
gata_1:
    puts   mess_no_file          ; Nu există nici un fișier cu numele dat
    puts   gen_name

```

```
        puts  <cr, lf>
gata:
        exit_dos          ; Ieșire în DOS
end start
```

Programul accesează PSP-ul, copiind "coada" comenzii într-un buffer propriu. Se sare apoi peste spațiile albe de la începutul șirului și de elimină și spațiile albe de la sfârșitul șirului (ptin mutarea terminatorului mai la stânga).

Apoi se copiază șirul "curățat" în bufferul gen\_name și se apelează procedura count, care întoarce numărul de cuvinte din șir. Dacă numărul de parametri este diferit de 1, se afișează un mesaj de help. Altfel, se setează zona DTA pe o zonă proprie și se apelează Find First. La un offset specificat în zona DTA, se va găsi numele primului fișier (dacă CF = 0).

Se deschide fișierul pentru citire și se intră într-o buclă de citire-afișare, care se încheie dacă numărul de octeți citați este strict mai mic decât cel cerut sau este zero.

Se apelează Find Next, cu parametrul zona proprie DTA, care întoarce numele următorului fișier identificat. Toată procedura se reia în buclă, până când Find Next raportează (prin CF = 0), că nu mai sunt fișiere care să corespundă numelui generic specificat.